NAG Library Routine Document

F07MSF (ZHETRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07MSF (ZHETRS) solves a complex Hermitian indefinite system of linear equations with multiple right-hand sides,

$$AX = B,$$

where A has been factorized by F07MRF (ZHETRF).

2 Specification

```
SUBROUTINE F07MSF (UPLO, N, NRHS, A, LDA, IPIV, B, LDB, INFO)
INTEGER               N, NRHS, LDA, IPIV(*), LDB, INFO
COMPLEX (KIND=nag_wp) A(LDA,*), B(LDB,*)
CHARACTER(1)          UPLO
```

The routine may be called by its LAPACK name ***zhetrs***.

3 Description

F07MSF (ZHETRS) is used to solve a complex Hermitian indefinite system of linear equations $AX = B$, this routine must be preceded by a call to F07MRF (ZHETRF) which computes the Bunch–Kaufman factorization of A.

If UPLO = 'U', $A = PUDU^{\mathrm{H}}P^{\mathrm{T}}$, where P is a permutation matrix, U is an upper triangular matrix and D is an Hermitian block diagonal matrix with 1 by 1 and 2 by 2 blocks; the solution X is computed by solving $PUDY = B$ and then $U^{\mathrm{H}}P^{\mathrm{T}}X = Y$.

If UPLO = 'L', $A = PLDL^{\mathrm{H}}P^{\mathrm{T}}$, where L is a lower triangular matrix; the solution X is computed by solving $PLDY = B$ and then $L^{\mathrm{H}}P^{\mathrm{T}}X = Y$.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

1: UPLO – CHARACTER(1) *Input*

On entry: specifies how A has been factorized.

UPLO = 'U'
: $A = PUDU^{\mathrm{H}}P^{\mathrm{T}}$, where U is upper triangular.

UPLO = 'L'
: $A = PLDL^{\mathrm{H}}P^{\mathrm{T}}$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\text{N} \geq 0$.

3: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides.

Constraint: $\text{NRHS} \geq 0$.

4: $\text{A}(\text{LDA}, *)$ – COMPLEX (KIND=nag_wp) array *Input*

Note: the second dimension of the array A must be at least $\max(1, \text{N})$.

On entry: details of the factorization of A, as returned by F07MRF (ZHETRF).

5: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F07MSF (ZHETRS) is called.

Constraint: $\text{LDA} \geq \max(1, \text{N})$.

6: $\text{IPIV}(*)$ – INTEGER array *Input*

Note: the dimension of the array IPIV must be at least $\max(1, \text{N})$.

On entry: details of the interchanges and the block structure of D, as returned by F07MRF (ZHETRF).

7: $\text{B}(\text{LDB}, *)$ – COMPLEX (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \text{NRHS})$.

On entry: the n by r right-hand side matrix B.

On exit: the n by r solution matrix X.

8: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07MSF (ZHETRS) is called.

Constraint: $\text{LDB} \geq \max(1, \text{N})$.

9: INFO – INTEGER *Output*

On exit: $\text{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

$\text{INFO} < 0$

If $\text{INFO} = -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

For each right-hand side vector b, the computed solution x is the exact solution of a perturbed system of equations $(A + E)x = b$, where

if UPLO = 'U', $|E| \leq c(n)\epsilon P|U||D||U^{\text{H}}|P^{\text{T}}$;

if UPLO = 'L', $|E| \leq c(n)\epsilon P|L||D||L^{\text{H}}|P^{\text{T}}$,

$c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

If $\hat{x}$ is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|x-\hat{x}\|_\infty}{\|x\|_\infty} \leq c(n)\,\mathrm{cond}(A,x)\epsilon$$

where $\mathrm{cond}(A,x) = \left\| |A^{-1}||A||x| \right\|_\infty / \|x\|_\infty \leq \mathrm{cond}(A) = \left\| |A^{-1}||A| \right\|_\infty \leq \kappa_\infty(A)$.

Note that $\mathrm{cond}(A,x)$ can be much smaller than $\mathrm{cond}(A)$.

Forward and backward error bounds can be computed by calling F07MVF (ZHERFS), and an estimate for $\kappa_\infty(A)$ ($= \kappa_1(A)$) can be obtained by calling F07MUF (ZHECON).

8 Parallelism and Performance

F07MSF (ZHETRS) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of real floating-point operations is approximately $8n^2r$.

This routine may be followed by a call to F07MVF (ZHERFS) to refine the solution and return an error estimate.

The real analogue of this routine is F07MEF (DSYTRS).

10 Example

This example solves the system of equations $AX = B$, where

$$A = \begin{pmatrix} -1.36+0.00i & 1.58+0.90i & 2.21-0.21i & 3.91+1.50i \\ 1.58-0.90i & -8.87+0.00i & -1.84-0.03i & -1.78+1.18i \\ 2.21+0.21i & -1.84+0.03i & -4.63+0.00i & 0.11+0.11i \\ 3.91-1.50i & -1.78-1.18i & 0.11-0.11i & -1.84+0.00i \end{pmatrix}$$

and

$$B = \begin{pmatrix} 7.79+\ \ 5.48i & -35.39+18.01i \\ -0.77-16.05i & 4.23-70.02i \\ -9.58+\ \ 3.88i & -24.79-\ \ 8.40i \\ 2.98-10.18i & 28.68-39.89i \end{pmatrix}.$$

Here A is Hermitian indefinite and must first be factorized by F07MRF (ZHETRF).

10.1 Program Text

```
    Program f07msfe

!     F07MSF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, x04dbf, zhetrf, zhetrs
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
```

```
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, lda, ldb, lwork, n,  &
                                          nrhs
      Character (1)                    :: uplo
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:,:), b(:,:), work(:)
      Integer, Allocatable             :: ipiv(:)
      Character (1)                    :: clabs(1), rlabs(1)
!     .. Executable Statements ..
      Write (nout,*) 'F07MSF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, nrhs
      lda = n
      ldb = n
      lwork = 64*n
      Allocate (a(lda,n),b(ldb,nrhs),work(lwork),ipiv(n))

!     Read A and B from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Read (nin,*)(a(i,i:n),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)(a(i,1:i),i=1,n)
      End If
      Read (nin,*)(b(i,1:nrhs),i=1,n)

!     Factorize A
!     The NAG name equivalent of zhetrf is f07mrf
      Call zhetrf(uplo,n,a,lda,ipiv,work,lwork,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!       Compute solution
!       The NAG name equivalent of zhetrs is f07msf
        Call zhetrs(uplo,n,nrhs,a,lda,ipiv,b,ldb,info)

!       Print solution

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04dbf('General',' ',n,nrhs,b,ldb,'Bracketed','F7.4',             &
          'Solution(s)','Integer',rlabs,'Integer',clabs,80,0,ifail)

      Else
        Write (nout,*) 'The factor D is singular'
      End If

    End Program f07msfe
```

10.2 Program Data

```
F07MSF Example Program Data
  4  2                                                     :Values of N and NRHS
  'L'                                                      :Value of UPLO
 (-1.36, 0.00)
 ( 1.58,-0.90) (-8.87, 0.00)
 ( 2.21, 0.21) (-1.84, 0.03) (-4.63, 0.00)
 ( 3.91,-1.50) (-1.78,-1.18) ( 0.11,-0.11) (-1.84, 0.00)  :End of matrix A
 ( 7.79,  5.48) (-35.39, 18.01)
 (-0.77,-16.05) (  4.23,-70.02)
 (-9.58,  3.88) (-24.79, -8.40)
 ( 2.98,-10.18) ( 28.68,-39.89)                            :End of matrix B
```

10.3 Program Results

```
F07MSF Example Program Results

Solution(s)
                    1                  2
1  ( 1.0000,-1.0000) ( 3.0000,-4.0000)
2  (-1.0000, 2.0000) (-1.0000, 5.0000)
3  ( 3.0000,-2.0000) ( 7.0000,-2.0000)
4  ( 2.0000, 1.0000) (-8.0000, 6.0000)
```